

Арьен Маркус

СОВРЕМЕННЫЙ FORTRAN

на практике

Арьен Маркус

Современный FORTRAN на практике

Modern Fortran in Practice

ARJEN MARKUS

WITH FOREWORD BY MICHAEL METCALF



CAMBRIDGE
UNIVERSITY PRESS

Современный Fortran на практике

АРЬЕН МАРКУС
С ПРЕДИСЛОВИЕМ МАЙКЛА МЕТКАЛФА

2-е издание, электронное



Москва, 2023

УДК 004.438Fortran

ББК 32.973.22

M27

Маркус, Аръен.

M27 Современный Fortran на практике / А. Маркус ; пер. с англ. А. В. Снастина. — 2-е изд., эл. — 1 файл pdf : 319 с. — Москва : ДМК Пресс, 2023. — Систем. требования: Adobe Reader XI либо Adobe Digital Editions 4.5 ; экран 10". — Текст : электронный.

ISBN 978-5-89818-602-9

Язык программирования Fortran изначально был предназначен для математических вычислений с максимальной производительностью. В последний стандарт Fortran 2008 включено множество современных функциональных возможностей: средства объектно-ориентированного программирования, специализированные операции с массивами, типы, определяемые пользователем и поддержка параллельных вычислений.

Данное учебное руководство поможет программистам на языке Fortran научиться применять все вышеперечисленные функциональные возможности в соответствии с современными требованиями: модульность, лаконичность, объектно-ориентированный подход и рациональное использование ресурсов, а также организация работы с учётом наличия нескольких процессоров. В книге рассматриваются практические примеры взаимодействия с программами, написанными на языке C, управления памятью, применения графики и графических пользовательских интерфейсов, параллельные вычисления с использованием библиотек MPI, OpenMP и комассивов (coarrays). Кроме того, автор анализирует некоторые числовые алгоритмы и их реализации, а также показывает, как можно применить некоторые библиотеки с открытыми исходными кодами.

УДК 004.438Fortran

ББК 32.973.22

Электронное издание на основе печатного издания: Современный Fortran на практике / А. Маркус ; пер. с англ. А. В. Снастина. — Москва : ДМК Пресс, 2015. — 318 с. — ISBN 978-5-97060-302-4. — Текст : непосредственный.

Все права защищены. Любая часть этой книги не может быть воспроизведена в какой бы то ни было форме и какими бы то ни было средствами без письменного разрешения владельцев авторских прав.

Материал, изложенный в данной книге, многократно проверен. Но поскольку вероятность технических ошибок все равно существует, издательство не может гарантировать абсолютную точность и правильность приводимых сведений. В связи с этим издательство не несет ответственности за возможные ошибки, связанные с использованием книги.

В соответствии со ст. 1299 и 1301 ГК РФ при устранении ограничений, установленных техническими средствами защиты авторских прав, правообладатель вправе требовать от нарушителя возмещения убытков или выплаты компенсации.

ISBN 978-5-89818-602-9

© Arjen Markus, 2012

© Оформление, перевод на русский язык
ДМК Пресс, 2015

«Eadem mutata resurgo»

(вольный перевод: «*изменяясь, я вновь воскресаю*»)

Надпись на могиле Якоба Бернулли,
изначально относящаяся к логарифмической спирали

Посвящается памяти моего отца.

*Мои родители научили меня любознательности.
Моя жена и мои дети всё время учат меня другим
важным вещам.*



ОГЛАВЛЕНИЕ

Предисловие Майкла Меткалфа.....	11
Предисловие автора	15
Глава 1. Введение в современный Fortran	17
1.1. Особенности современного Fortran	17
1.2. Fortran 90	21
1.3. Fortran 95	25
1.4. Fortran 2003	26
1.5. Fortran 2008	28
1.6. Что осталось неизменным.....	29
Глава 2. Функции для работы с массивами	32
2.1. Передача массивов в аргументах.....	33
Производительность при использовании функций обработки массивов.....	35
2.2. Элементные функции и автоматическое перераспределение памяти	36
2.3. Два более сложных примера	38
Дистанционирование иррациональных чисел.....	38
Быстрая сортировка QuickSort.....	40
2.4. Компактный стиль.....	41
Глава 3. Математические абстракции	44
3.1. Автоматическое дифференцирование	44
Проблемы при вычислениях	49
3.2. Дискретное программирование	50
Управление памятью	51
3.3. Перечислимое множество решений Диофантовых уравнений	53
3.4. Отложенные или ленивые вычисления	56

Глава 4. Управление памятью	60
4.1. Динамически изменяемые массивы.....	60
4.2. Утечки памяти при использовании указателей	61
4.3. Увеличение размера массива	62
4.4. Строки символов с изменяемой длиной.....	63
4.5. Сочетание автоматических и динамических массивов	66
4.6. Производительность массивов разных типов.....	67
4.7. Параметризованные производные типы	69
4.8. Утечки памяти в производных типах	71
4.9. Производительность и доступ к памяти.....	76
Глава 5. Проблема интерфейса	80
5.1. Подстановка параметров	82
5.2. Использование пула данных.....	84
Данные в модулях.....	85
Внутренние подпрограммы	88
5.3. Передача дополнительных аргументов	89
Массив параметров.....	89
Использование функции transfer()	90
Процедуры, связанные с типом	91
Указатели на процедуры.....	93
5.4. Управляющие конструкции	95
Библиотека OpenMP	100
5.5. Работа с числовыми значениями различной точности	102
5.6. Резюме	103
Глава 6. Взаимодействие с программами на языке С на примере работы с СУБД SQLite	105
6.1. Соответствие типов данных	106
6.2. Передача аргументов между подпрограммами, написанными на С и на Fortran.....	109
6.3. Соглашения об именовании и вызовах функций	110
6.4. Работа с производными типами	113
6.5. Создание интерфейса к СУБД SQLite	116
Глава 7. Графика, GUI и Интернет	124
7.1. Вывод результатов в графическом виде	125
7.2. Графические пользовательские интерфейсы (GUI)	131

7.3. Интернет.....	139
7.4. Работа с XML-файлами	143
Глава 8. Модульное тестирование	148
8.1. Инструментальные средства тестирования.....	148
8.2. Пример: обработка трёхдиагональной матрицы.....	149
8.3. Проектирование и реализация.....	153
8.4. Заключительные замечания	155
Глава 9. Просмотр и рецензирование исходного кода	157
9.1. Соблюдать определённости и однозначности	158
Используйте явные объявления переменных и констант.....	158
Используйте предусловия	160
Переменные, сохраняющие свои значения между вызовами.....	161
Видимость интерфейса к подпрограмме или функции.....	161
Доступность переменных и подпрограмм	161
Вариант default в блоке select и ветвь else в блоке if.....	162
Информативные сообщения об ошибках	162
9.2. Избегать излишней сложности и запутанности	164
9.3. Избегать «ловушек»	169
Правильная обработка ошибок	169
Сравнение вещественных чисел	169
Смешанная точность	171
Неожиданные результаты при работе с отрицательными числами.....	171
Автоматические массивы	172
Ошибки могут возникать не только при работе с числами.....	173
9.4. Писать простой и понятный код	173
Глава 10. Устойчивая к ошибкам реализация нескольких простых алгоритмов	178
10.1. Обзор существующих подобных методик.....	179
10.2. Линейная интерполяция.....	181
10.3. Простые статистические методы и характеристики.....	187
10.4. Поиск корней уравнения	195
Глава 11. Объектно-ориентированное программирование	209
11.1. Расширение типов и процедуры, связанные с типами... ..	209
Передача объекта в другом аргументе.....	211

Расширение до трёх измерений	212
Пример: случайные перемещения в двух и в трёх измерениях.....	215
Определение динамического типа.....	217
Наблюдение за частицами.....	217
11.2. Интерфейсы как контракты	221
Аппроксимация множественного наследования	225
11.3. Использование прототипирования.....	226
Пример: моделирование поведения рыб.....	229
11.4. Абстрактные типы данных и обобщённое программирование	232
11.5. Изменение поведения типа данных.....	235
11.6. Шаблоны проектирования.....	237
Шаблон проектирования Factory.....	238
Шаблон проектирования Наблюдатель.....	241
Глава 12. Параллельное программирование	245
12.1. Простые числа	246
Библиотека OpenMP	248
Интерфейс MPI.....	251
Комассивы	255
12.2. Декомпозиция по доменам	259
OpenMP.....	261
MPI.....	265
Комассивы	266
12.3. Другие методики параллельного программирования....	268
12.4. Резюме	270
Приложение А. Инструментальные средства для разработки и сопровождения	271
А.1. Компиляторы	271
А.2. Средства сборки программ	272
А.3. Интегрированные среды разработки	275
А.4. Средства проверки во время выполнения.....	276
А.5. Системы управления версиями	278
А.6. Документирование исходного кода.....	279
А.7. Охват кода тестированием и статический анализ.....	281
Приложение Б. Некоторые нюансы использования Fortran.....	285

Б.1. Особенности стандарта	285
Вычисление логических выражений по короткой схеме	285
Сохранение значений локальных переменных	286
Ещё об инициализации	287
Двойная точность и вычисление правой части выражений	287
Передача одного и того же аргумента дважды	288
REAL(4)	290
Признак конца файла (EOF), вывод на экран и т. п.	290
Внешние и внутренние (встроенные) подпрограммы	291
Несовпадения в интерфейсах: предполагаемая форма и явная форма массивов	292
Инициализация генератора случайных чисел	293
Открытие одного и того же файла дважды	294
Б.2. Массивы	294
Использование автоматических и временных массивов может привести к переполнению стека	294
Границы массивов с начальным индексом меньше 1	296
Объявления массивов: dimension(:) и dimension(*)	296
Б.3. Динамические библиотеки	297
Открытие файла в программе и использование его в DLL и наоборот	297
Выделение памяти в DLL и освобождение этой памяти в программе и наоборот	298
Аргументы командной строки недоступны в DLL	298
Подпрограммы или данные из основной программы, используемые в DLL	298
Приложение В. Зарегистрированные товарные знаки, упоминаемые в данной книге	300
Список литературы	302
Предметный указатель	309



ПРЕДИСЛОВИЕ

МАЙКЛА МЕТКАЛФА

Эпоха, когда прикладные программы создавались на языке Fortran, почти полностью совпадает с периодом существования компьютеров общего назначения. Это удивительный факт, и с учётом того, что многие другие языки программирования высокого уровня прекратили своё существование, трудно понять, почему получилось именно так. Возможно, исходные принципы проектирования Джона Бэкуса (John Backus) – простота использования и эффективность выполнения – стали двумя решающими факторами. Возможно, сыграла роль преданность языку Fortran сообщества его пользователей, которые всегда старались не отставать от новейших разработок в области технологии программирования и адаптировать язык к постоянно расширяющемуся кругу требований.

В течение нескольких десятилетий Fortran считался вымирающим языком, но, несмотря на все предсказания, оказался на удивление живучим. Более того, в последние годы возобновилась его стандартизация, и последний стандарт Fortran 2008 должен снова продлить жизнь этому языку. С учётом этих нововведений очень жаль, что продолжают существовать старые версии Fortran, как в форме давно устаревших курсов, читаемых неисправимо упрямыми преподавателями, так и в виде вышедших из употребления концепций, о которых постоянно твердят его критики. Современный Fortran – это процедурный, императивный, компилируемый язык с синтаксисом, соответствующим точному представлению математических формул. Независимые процедуры могут компилироваться отдельно или объединяться в модули, что упрощает создание крупномасштабных программ и библиотек процедур. В язык включены функциональные возможности для обработки массивов, абстрактные типы данных, динамические структуры данных, средства объектно-ориентированного программирования и параллельной обработки. Fortran способен без затруднений взаимодействовать с C. Таким образом, современный

Fortran, начиная с версии Fortran 95 (так теперь стали обозначаться версии стандарта) – это мощный инструмент. Он в полной мере поддерживает структурное программирование, а средства объектно-ориентированного программирования, появившиеся в стандарте Fortran 2003, стали самым значительным усовершенствованием языка, его главным нововведением. Большинство из упомянутых новых функциональных возможностей описано в данной книге.

Но ни один стандарт до Fortran 2003 включительно не содержал никаких средств, специально предназначенных для поддержки параллельного программирования. Такая поддержка осуществлялась опосредованно, с привлечением вспомогательных стандартов, в частности HPF (High-Performance Fortran), MPI (Message Passing Interface), OpenMP и Posix Threads (Pthreads). Использование библиотек MPI и OpenMP стало массовым явлением, но HPF в конечном счёте не имел особого успеха. Сейчас, после принятия стандарта Fortran 2008 одним из самых сильных свойств современного языка Fortran является непосредственная поддержка параллельного программирования, благодаря введению чрезвычайно востребованного средства: комассивов (coarrays).

Директивы HPF имели форму строк комментариев и распознавались только HPF-процессором. Например, существовала возможность выравнивания трёх совпадающих по форме массивов по четвёртому с обеспечением локальности ссылок. Другие директивы позволяли распределить обработку выравниваемых массивов по нескольким процессорам. С другой стороны, MPI представляет собой универсальную библиотеку процедур для передачи сообщений, а библиотека OpenMP, поддерживающая независимое от платформы параллельное программирование с совместным использованием памяти, состоит из набора директив компилятора, библиотечных подпрограмм и переменных среды, которые определяют поведение программы во время выполнения. Posix Threads – это стандарт, определяющий спецификацию библиотеки для поддержки многопоточности.

В отличие от всех перечисленных средств, главной целью введения комассивов является предоставление синтаксиса, минимально влияющего на внешний вид программы и позволяющего распределить по нескольким процессорам не только данные, как в модели «одна инструкция, много данных» (Single Instruction Multiple Data, SIMD), но и работу в соответствии с моделью «одна программа, много данных» (Single Program Multiple Data, SPMD). От программиста требу-

ется знание лишь небольшого набора новых правил. Работа с комассивами – это самое важное новшество в стандарте Fortran 2008, но кроме него была введена новая форма управляющей конструкции `do concurrent` как способ распараллеливания циклов. Вполне очевидно, что появилась возможность обеспечения полноценного режима параллельного выполнения, не выходя за рамки языка. Всё это также рассматривается и сравнивается в данной книге.

Другие важные нововведения в стандарте Fortran 2008: подмодули (submodules), более удобный доступ к объектам данных, усовершенствованные средства ввода/вывода и управления выполнением, дополнительные внутренние процедуры, в частности, для работы с битами. Fortran 2008 был опубликован в 2010 году, и в настоящее время является действующим стандартом. Будущее языка Fortran определяет его способность обеспечить высокую производительность вычислений, таким образом, комассивы становятся важнейшим инструментом языка.

Но языку программирования трудно выжить, если о нём мало что известно. Причём должны существовать не только учебники по его синтаксису и семантике, но и книги о практическом применении языка для решения реальных задач. Опыт работы, конкретные методики, а также способы наиболее оптимального использования новых функциональных возможностей должны передаваться новому поколению программистов. В наше время, когда языки программирования не являются «вещью в себе», а всё чаще используются совместно друг с другом или в сочетании с разнообразными инструментальными средствами, необходима именно такая книга как «Современный Fortran на практике».

Автор этой книги постоянно сотрудничает с информационным бюллетенем ACM Fortran Forum и является активным участником группы новостей `comp.lang.fortran`, где публикует множество полезных советов. Его опыт научного программирования приносит пользу сообществу не только в Нидерландах, где он проживает, но и во всём мире, а статьи по обобщённому программированию и использованию шаблонов проектирования в Fortran содержат много свежих идей. Таким образом, квалификация автора вполне позволяет ему написать книгу, подобную этой.

Но «Современный Fortran на практике» – это не просто сборник предыдущих публикаций. Она содержит логически связанное изложение основ, и собственные материалы автора по параллельному программированию на Fortran с использованием библиотек MPI,

OpenMP и комассивов (в кратком изложении), а также описывает использование Fortran для создания графических приложений и для взаимодействия с графическими пользовательскими интерфейсами. В книге много подробных примеров кода, на которые читатель может опираться при создании собственных программ.

Эта книга весьма полезна, и я рекомендую её всем программистам, использующим Fortran. Все мы уже убедились в том, что Fortran продолжает существовать и у него есть будущее.

Майкл Меткалф (Michael Metcalf)
Токио, октябрь 2011 г.



ПРЕДИСЛОВИЕ АВТОРА

Я программирую на языке Fortran уже более 25 лет, сначала на FORTRAN IV, немного позже на FORTRAN 77. В последнее десятилетие прошлого века я и несколько моих коллег прослушали курс по Fortran 90, прочитанный Яном ван Оостервийком (Jan van Oosterwijk) в Техническом университете Делфта. Приблизительно в это же время я присоединился к группе новостей comp.lang.fortran, и многому научился в этом дружелюбном сообществе.

В определённом смысле я обычный программист на языке Fortran. Моя основная специальность – физика, а программирование я начал осваивать ещё во время учёбы, но большую часть практического опыта приобрёл во время работы. Я стал программистом по производственной необходимости, а не из-за особого интереса к всё более изощрённым возможностям программирования вообще и к перспективам их применения в Fortran. Позже я начал писать статьи для информационного бюллетеня ACM Fortran Forum, которые стали отправной точкой для данной книги.

Эта книга не научит вас программированию на языке Fortran. Для этого существует множество специальных учебников ([22], [65]). Цель данной книги – показать, как можно использовать современный Fortran для решения задач, существующих в настоящее время, продемонстрировать, например, что методики, широко распространённые в мире объектно-ориентированных языков, таких как Java и C++, вполне применимы в последней версии Fortran. Более того, в книге описываются некоторые приемы решения задач программирования, которые не так-то просто реализовать с помощью других языков.

Если вы знакомы с языком в основном по старым его версиям, существовавшим до появления Fortran 90, несколько первых глав постепенно познакомят вас с операциями над массивами, с перегружаемыми операциями и с некоторыми другими функциональными возможностями, введёнными в этом стандарте. Кроме того, вы увидите, что при использовании Fortran появилась возможность приме-

нения совершенно различных стилей программирования, присущих в основном *функциональным* языкам программирования. В большинстве глав показано, как применять на практике все эти возможности языка.

В данной книге часто встречаются ссылки на программы, которые я написал и опубликовал на веб-сайте SourceForge, или на программы, к созданию которых я причастен в той или иной степени. Не считайте это рекламой конкретного программного обеспечения или навязыванием своего мнения – это просто моя уверенность в хорошем качестве упоминаемых программ. Во всех примерах, код которых был написан не мной, я старался указывать авторов, но я всего лишь обычный человек, и мог забыть пару имён.

Книги, подобные этой, практически невозможно написать в полной изоляции. Я благодарен Майклу Меткалфу (Michael Metcalf) и Яну Чиверсу (Ian Chivers), редакторам бюллетеня ACM Fortran Forum, всем участникам группы новостей comp.lang.fortran, а также Билу Клебу (Bil Kleb), Полу ван Делсту (Paul van Delst), Рольфу Аде (Rolf Ade), Генри Гарднеру (Henry Gardner), Саймону Гирду (Simon Geard), Рихарду Захенвирту (Richard Suchenwirth), Дэниелу Крафту (Daniel Kraft), Рихарду Майне (Richard Maine), Стиву Лайонелу (Steve Lionel), Кэмерону Лэйрду (Cameron Laird) и Клифу Флинту (Clif Flynt). Спасибо им за обсуждения и обзоры моих работ, за дискуссии по различным аспектам программирования на языке Fortran и программирования в целом, за уделённое мне внимание.

У этой книги есть свой веб-сайт http://flibs.sf.net/examples/modern_fortran.html, где размещены полные исходные коды всех примеров. Я протестировал их с помощью компиляторов *gfortran* и *Intel Fortran* главным образом в ОС Windows, но проверял и в Linux. Некоторые из этих программ используют самые последние нововведения в стандарте Fortran, поэтому вам потребуется наиболее свежая версия компилятора.

Ариен Маркус (Arjen Markus)
Роттердам, ноябрь 2011 г.



ГЛАВА 1.

Введение в современный Fortran

С момента публикации стандарта FORTRAN 77 в 1978 году язык программирования Fortran претерпел множество изменений [61].¹ Вносимые изменения учитывали как новые подходы к методикам программирования, так и новые разработки в области компьютерной аппаратуры. Язык изначально был задуман как максимально эффективный инструмент вычислений. В самом последнем на момент написания этой книги стандарте, Fortran 2008, введена прямая поддержка параллельной обработки средствами самого языка, то есть основная идея эффективности получила дальнейшее развитие [71].

В этой главе приводится общий обзор различных стандартов, появившихся после FORTRAN 77. Не следует считать это попыткой полного описания всех вводимых стандартами новых функциональных возможностей, для этого потребовалась бы отдельная книга или даже несколько книг. Более подробные описания версий стандартов можно найти в книгах Меткалфа (Metcalf) [63], [65] или Брэйнерда (Brainerd) и др. [36].

1.1. Особенности современного Fortran

Стандарт Fortran 90 ввёл весьма существенные изменения по сравнению с повсеместно распространённым стандартом FORTRAN 77: свободный формат исходного кода, операции с массивами, модули, производные типы и т. п. Чтобы понять, что это значило для програм-

¹ Формально в официальных документах наименование этой версии стандарта должно быть записано прописными буквами: FORTRAN 77. Начиная с версии стандарта Fortran 90, наименование языка пишется строчными буквами (буквами в нижнем регистре).

миста, рассмотрим простую задачу: имеется файл с набором чисел, в каждой строке содержится одно значение (для простоты), необходимо построить простую гистограмму, показывающую распределение этих чисел. Программа, решающая такую задачу на FORTRAN 77, могла выглядеть следующим образом:²

```

*
* Построение простой гистограммы.
*

      PROGRAM HIST

      INTEGER MAXDATA
      PARAMETER (MAXDATA = 1000)
      INTEGER NOBND
      PARAMETER (NOBND = 9)
      REAL BOUND(NOBND)
      REAL DATA(MAXDATA)
      INTEGER I, NODATA

      DATA BOUND /0.1, 0.3, 1.0, 3.0, 10.0, 30.0,
&            100.0, 300.0, 1000.0/

      OPEN( 10, FILE = 'histogram.data', STATUS = 'OLD', ERR = 900 )
      OPEN( 20, FILE = 'histogram.out' )

      DO 110 I = 1,MAXDATA
      READ( 10, *, END = 120, ERR = 900 ) DATA(I)
110  CONTINUE
*
120  CONTINUE
      CLOSE( 10 )
      NODATA = I - 1

      CALL PRHIST( DATA, NODATA, BOUND, NOBND )
      STOP

*
* Файл не найден и другие ошибки.
*
900  CONTINUE
      WRITE( *, * ) 'File histogram.data could not be opened'
&                'or some reading error'
      END

*
* Подпрограмма для вывода гистограммы.
*

      SUBROUTINE PRHIST( DATA, NODATA, BOUND, NOBND )

```

² Данный листинг демонстрирует жёстко фиксированный формат FORTRAN 77: позиции 1–5 – метка, позиция 6 – признак продолжения строки, позиции 7–72 – оператор. Именно в этом и состоит главная задача этого листинга. – *Прим. перев.*

```

      REAL DATA(*), BOUND(*)
      INTEGER NODATA, NOBND

      INTEGER I, J, NOHIST

      DO 120 I = 1,NOBND
      NOHIST = 0
      DO 110 J = 1,NODATA
          IF ( DATA(J) .LE. BOUND(I) ) THEN
              NOHIST = NOHIST + 1
          ENDIF
      110CONTINUE
      WRITE( 20, '(F10.2,I10)' ) BOUND(I), NOHIST
      120 CONTINUE
      END

```

На Fortran 90 ту же программу можно переписать в так называемом *свободном формате (free form)* с использованием различных функций запросов и операций над массивами:

```

! Построение простой гистограммы.
!
program hist
  implicit none

  integer, parameter :: maxdata = 1000
  integer, parameter :: nobnd = 9

  real, dimension(maxdata) :: data
  real, dimension(nobnd) :: &
    bound = (/0.1, 0.3, 1.0, 3.0, 10.0, &
      30.0, 100.0, 300.0, 1000.0/)

  integer :: i, nodata, ierr

  open( 10, file = 'histogram.data', status = 'old', &
    iostat = ierr )

  if ( ierr /= 0 ) then
    write( *, * ) 'file histogram.data could not be opened'
    stop
  endif

  open( 20, file = 'histogram.out' )

  do i = 1,size(data)
    read( 10, *, iostat = ierr ) data(i)

    if ( ierr > 0 ) then
      write( *, * ) 'Error reading the data!'
    endif
  enddo

```

```

                                stop
                    elseif ( ierr < 0 ) then
                                exit ! Обнаружен признак конца файла.
                    endif
            enddo

            close( 10 )
            nodata = i - 1
            call print_history( data(1:nodata), bound )

contains

! Подпрограмма для вывода гистограммы.
!
subroutine print_history( data, bound )
    real, dimension(:), intent(in) :: data, bound

    integer :: i

    do i = 1, size(bound)
        write( 20, '(f10.2,i10)' ) &
            bound(i), count( data <= bound(i) )
    enddo
end subroutine print_history

end program hist

```

Основные различия между программами:

- стандарт Fortran 90 и все последующие стандарты официально разрешали записывать исходный код программы в свободном формате с использованием букв нижнего регистра, хотя и до этого многие компиляторы, соответствующие стандарту FORTRAN 77, допускали такую запись в качестве неофициального расширения;
- новый оператор `implicit none` заставляет компилятор следить за тем, чтобы все переменные объявлялись с явно указанным типом. Это позволяет устранить большую группу потенциальных ошибок;
- использование *внутренней подпрограммы (internal routine)*, обозначенной ключевым словом `contains`, позволяет компилятору проверить правильность её действительных аргументов, соответствие их количества и типов.³
- исключена необходимость использования оператора `GOTO`,

³ Это только одно из множества полезных свойств внутренних подпрограмм, подробнее см. раздел 1.2.

- имеющего дурную репутацию, то есть его можно заменить «более структурным» аналогом `exit` для завершения цикла `do`;
- появилась возможность использования *сечений массивов* (*array sections*), например, `data(1:nodata)`, для передачи только требуемой части массива `data`, а также *функции запроса* (*inquiry function*) `size()`, возвращающей соответствующее количество элементов массива. Кроме всего прочего это сделало ненужными два аргумента подпрограммы, в которых передавались размеры массивов;
 - использование стандартной функции `count()` позволило убрать полностью цикл `do` при определении данных гистотграммы.

Тем не менее, следует отметить, что первая программа остаётся абсолютно корректной даже с точки зрения современных стандартов Fortran, что является очень важным аспектом использования данного языка. Это означает, что вы можете постепенно переходить к применению новых функциональных возможностей, а не переписывать всю программу целиком.

1.2. Fortran 90

В стандарт Fortran 90 было введено большое число новых функциональных свойств, из которых наиболее известными стали операции с массивами. Но стандарт определяет и другие важные нововведения:

- оператор `implicit none` требует от пользователя явно объявлять все переменные, а компилятор обязан проверять их. Это позволяет избавиться от опечаток в именах переменных;
- *модули* (*modules*) в сочетании с операторами или атрибутами `public` и `private` представляют собой эффективные средства деления больших программ на более мелкие части, режим доступа к которым определяется явно в исходном коде. Кроме того, модули позволяют формировать чётко определённые интерфейсы к содержащимся в них подпрограммам, что даёт возможность компилятору выполнить весь комплекс проверок и оптимизаций. Если это недоступно, например, при взаимодействии с подпрограммами на другом языке, можно воспользоваться *интерфейсными блоками* (*interface blocks*);
- не только основная программа, но также подпрограммы и функции могут содержать так называемые *внутренние под-*

программы (internal routines). В такой подпрограмме доступны все переменные из внешней содержащей её (под)программы, но кроме того внутренняя подпрограмма создаёт новую *область видимости (scope)*, где определяются локальные переменные. Это дополнительный способ деления кода на модули;

- современные компьютеры предоставляют разнообразные средства управления памятью, которые не были так широко распространены во времена стандарта FORTRAN 77, поэтому некоторые положения стандарта Fortran 90 связаны с управлением памятью:
 - разрешены *рекурсивные подпрограммы (recursive routines)*, упрощающие реализацию рекурсивных алгоритмов;
 - с помощью операторов `allocate` и `deallocate` программист может вручную регулировать размер массивов в соответствии с условиями решаемой задачи. Массивы, размер которых можно регулировать вручную, делятся на две категории: динамические (`allocatable`) и статические (`pointer`). Первые предлагают больше возможностей для оптимизации, тогда как вторые дают большую гибкость;
 - помимо прямого изменения размеров массива программист может использовать *автоматические массивы (automatic arrays)* – массивы, которые получают свой размер из формальных аргументов, автоматически создаются при входе в подпрограмму или функцию и автоматически удаляются при выходе;
- *операции с массивами (array operations)* – одно из важнейших нововведений в стандарте Fortran 90, поскольку поощряют лаконичный стиль программирования и существенно упрощают задачу оптимизации для компилятора. Эти операции поддерживаются не только в обычных арифметических выражениях, но и с помощью набора обобщённых стандартных функций.

Можно работать как с целым массивом, так и с некоторой его частью, определяемой начальной и конечной позициями, и *шагом по индексу (stride)* в любом измерении.

Кроме того, функции, определённые пользователем, могут возвращать массив как результат, расширяя возможности операций с массивами;

- к традиционному старому *фиксированному формату (fixed form)*, присущему языку Fortran с момента его создания, офи-

циально добавлен *свободный формат (free form)*, в котором столбцы 1–6 уже не имеют какого-либо специального предназначения. Свободный формат придаёт исходному коду более современный вид.

- массивы значений можно формировать «на лету» с помощью так называемых *конструкторов массивов (array constructors)*. Это мощный механизм, удобный для заполнения массивов значениями;
- как и многие современные языки программирования Fortran поддерживает определение новых структур данных. Для этого предназначен механизм *производных типов (derived types)*. Любой производный тип состоит из элементов разных типов – простых, таких как `integer` или `real`, или других производных типов. Производные типы можно применять точно так же как простые – передавать их в подпрограммы, использовать в качестве возвращаемых значений или создавать массивы этих типов.

Кроме того, производные типы можно использовать для создания связанных списков, деревьев и других известных абстрактных типов данных;

- *перегрузка (overloading)* подпрограмм и операций, таких как сложение и вычитание, предоставляет возможность расширения языка новыми полнофункциональными типами. Программист может определить обобщённые имена для специализированных подпрограмм и/или переопределить операции `+`, `-` и другие для числовых типов, не регламентируемых стандартом (например, для рациональных чисел).

Фактически это дает возможность определять собственные операции. Простой пример: предположим, что программа работает с плоскими геометрическими объектами, поэтому имеет смысл определить операцию *пересечения (intersection)* для двух таких объектов `object_a.intersects(object_b)`, чтобы заменить вызов функции `intersect_objects(object_a, object_b)`;

- теперь функции и подпрограммы могут иметь *необязательные аргументы (optional arguments)*, при этом функция `present()` позволяет определить наличие или отсутствие аргумента. Также можно вызывать функции и подпрограммы с аргументами, указанными в произвольном порядке, при условии передачи имен формальных аргументов, чтобы компилятор мог приве-

сти в соответствие фактические аргументы с формальными. Ещё одно усовершенствование – можно задать *назначение* (*intent*) любого аргумента: только для ввода, только для вывода или для ввода/вывода. Это дополнительное средство документирования программы, а также информация для выполнения компилятором оптимизации определённого типа;

- *разновидности типов* (*kinds*) – это специфический для языка Fortran способ определения характеристик простых типов. Например, для переменных типа `real` можно выбирать единичную или двойную точность не явным указанием типа (`real` или `double precision`), а с помощью ключевого слова `kind`:

```
integer, parameter :: double = &  
    select_kind_real( precision, range )  
real(kind=double)   :: value
```

- в дополнение к новой конструкции `select/case`, циклам `do while` и циклам без условия `do`, появилась возможность давать имена всем управляющим структурам. Это упрощает документирование начала и конца таких структур. Чтобы пропустить некоторую часть тела цикла `do`, теперь можно воспользоваться оператором `cycle` (с необязательным именем конкретного цикла) – таким способом можно выйти из нескольких вложенных циклов `do`. Подобным образом оператор `exit` завершает выполнение цикла `do`;
- стандарт Fortran 90 определяет большое количество функций и подпрограмм:
 - числовые функции-запросы для получения свойств модели вещественных чисел;
 - функции обработки массивов, в большинстве случаев принимающие выражения с массивами, как один из аргументов. Например, подсчитать число положительных элементов в массиве можно так:

```
integer, dimension(100,100) :: array  
...  
write(*,*) 'Number of positive elements: ', &  
    count( array > 0 )
```

- функции для работы с символами и функции для операций с битами;

- к усовершенствованиям системы ввода/вывода относится не-продвигающий ввод/вывод (nonadvancing I/O), то есть программа может манипулировать не только целыми записями, но также читать или записывать часть записи в одном операторе, а остальную часть записи в другом.

1.3. Fortran 95

Стандарт Fortran 95 представляет собой скорректированную и дополненную версию предыдущего выпуска стандарта, поэтому различия не столь значительны. Тем не менее, следует обратить внимание на следующие важные изменения:

- согласно стандарту Fortran 90, переменные с атрибутом `pointer` не могут инициализироваться. Начальное состояние не определено ни для связанных, ни для несвязанных переменных. В версии Fortran 95 введена функция `null()` для явной инициализации указателей:

```
real, dimension(:), pointer :: ptr => null()
```

- динамические (`allocatable`) локальные переменные без атрибута `save` автоматически удаляются из памяти при выходе из области видимости (при возврате из подпрограммы или из функции). Это вполне безопасно, поскольку после выхода из области видимости таких переменных, выделенная им память становится недоступной;
- технический отчёт, как подготовительный материал для следующего стандарта, описывает, как динамически размещаемые (`allocatable`) компоненты могут входить в состав производных типов и возвращаться функциями. (Этот технический отчёт в дальнейшем стал частью стандарта Fortran 2003 с некоторыми дополнениями.);
- в Fortran 95 были введены типы подпрограмм `pure` и `elemental`. *Элементные подпрограммы (elemental routines)* освобождают программиста от необходимости писать несколько версий подпрограмм для работы с массивами всех требуемых размерностей. Многие функции и подпрограммы стандартной библиотеки уже имели статус `elemental`, означающий, что они работают с отдельными элементами массивов, которые могут передаваться без соблюдения какого-либо порядка.

После ввода стандарта Fortran 95 программистам была предоставлена возможность самим писать такие подпрограммы.

«Чистые» (pure) подпрограммы предоставляют компилятору больше возможностей для оптимизации, так как считается, что они не имеют побочных эффектов.

- следует особо отметить появление оператора forall, взятого из языка High Performance Fortran. Он был предназначен для расширения возможностей операций над массивами, но на практике далеко не всегда использовался корректно. (Стандартом Fortran 2008 введена более гибкая конструкция do concurrent.)

1.4. Fortran 2003

Fortran 2003 представляет собой существенно переработанную и исправленную версию стандарта, и самым главным изменением является введение поддержки объектно-ориентированного программирования.⁴ Но в этой версии появились и другие нововведения, например, стандартизация взаимодействия с подпрограммами (функциями), написанными на языке C. Ниже кратко описаны основные нововведения в стандарте Fortran 2003:

- с поддержкой *объектно-ориентированного программирования* (object-oriented programming) были напрямую связаны некоторые новые функциональные возможности:
 - производные типы теперь могут содержать процедуры (функции и подпрограммы), связанные с данным типом (так, что все переменные этого типа могут пользоваться одним и тем же набором процедур) или с конкретной переменной. Такие процедуры автоматически принимают «свою» переменную (или объект в более широком смысле) как один из аргументов, но при этом программисту предоставляется возможность управления передачей объектов. (Эти вопросы будут рассматриваться более подробно в главе 11.);

⁴ Стандарт Fortran 90 определяет достаточно много функциональных свойств языка, позволяющих программисту вплотную приблизиться к этому стилю программирования, но в нём отсутствует наследование, зачастую рассматриваемое как одна из наиболее важных характеристик объектно-ориентированного программирования. Поэтому Fortran 90 иногда называли языком, основанным на концепции объектов (object-based).



- появилась возможность на основе производных типов создавать новые типы, то есть «расширять» производные типы. Это механизм наследования в Fortran. *Расширенный тип (extended type)* может содержать новые компоненты (данные и процедуры), а также переопределять процедуры, связанные с родительским типом, но не имеет права изменять их сигнатуру;
- в оператор `select` добавлена возможность выбора по *типу* переменной. Это важно при работе с так называемыми *полиморфными переменными (polymorphic variables)* — переменными-указателями, тип которых меняется в зависимости от типов связанных с ними переменных в тот или иной момент выполнения программы. Для объявления полиморфных переменных используется ключевое слово `class` вместо `type`.⁵
- чтобы обеспечить большую гибкость механизма расширения типов, для процедур добавлена характеристика *абстрактный интерфейс (abstract interface)*. Вместо определения конкретной подпрограммы интерфейсы объявляют лишь «внешний вид» подпрограммы со списком аргументов и типом возвращаемого значения. В дальнейшем это объявление может служить шаблоном для настоящих подпрограмм;
- поскольку *указатели на процедуры (procedure pointers)* чаще встречаются в виде компонентов производных типов, их можно использовать как обычные переменные;
- для получения некоторых особых эффектов была введена концепция *внутренних модулей (intrinsic modules)*. Примером таких особых эффектов может служить управление моделью представления вещественных чисел: режим округления, эффект исключений, связанных с операциями над вещественными числами, а также организация взаимодействий с программами на языке C, так как при этом иногда требуется соблюдение разных соглашений об именовании и формате вызова функций;
- улучшено управление памятью:
 - длину строки символов теперь можно определить с помощью оператора `allocate`;

⁵ В [73] сравнивается терминология объектно-ориентированных концепций в языках Fortran и C++.

- при некоторых условиях динамический массив можно автоматически привести к корректному размеру;
- допускается перемещать выделенную память из одной переменной в другую с помощью подпрограммы `move_alloc()`. Это упрощает увеличение размера массива;
- появилась весьма полезная возможность *потокowego доступа* (*stream access*) к файлам, которая позволяет читать и записывать содержимое файлов не только целыми записями. В частности, теперь можно читать и записывать двоичные файлы, не имеющие внутренней структуры. Для старых файлов, *не имеющих формата*, (*unformatted files*) требовалась структура, основанная на записях, что затрудняло взаимодействие с другими языками программирования;
- стандарт Fortran 2003 также регламентирует доступ к системной среде в форме *переменных окружения* (*environment variables*) и к аргументам командной строки, заданным при запуске программы. Раньше для доступа к этой информации программист был вынужден использовать специфические возможности конкретного компилятора.

1.5. Fortran 2008

В версии стандарта Fortran 2008 самым важным нововведением являются *комассивы* (*coarrays*), но все прочие изменения языка были незначительными. Более подробно комассивы рассматриваются в главе 12 [72], [71]. Помимо этого в стандарте определено несколько новых конструкций и ключевых слов, а также новые стандартные функции:

- комассивы представляют собой механизм параллельных вычислений из категории «разделённое глобальное адресное пространство» (*Partitioned Global Address Space, PGAS*). По существу этот механизм обеспечивает доступность данных нескольким копиям программы и освобождает программиста от обязанности выбирать конкретный способ передачи данных. Компилятор должен сгенерировать код, выполняющий эту задачу рационально и эффективно;
- массивы могут быть определены как *непрерывные* (*contiguous*). Это позволяет компилятору повысить степень оптимизации кода, так как элементы массива в памяти следуют друг за другом;



- конструкция `block – end block` определяет локальную область видимости в программе или в подпрограмме, так что можно объявлять новые переменные, которые будут существовать только внутри этого блока;
- модули, введенные в стандарте Fortran 90, являются важным механизмом деления больших программ на части. Но сам по себе этот механизм не может обеспечить высокую степень модульности, поскольку компилироваться должен весь модуль, целиком. В стандарт Fortran 2003 были введены *подмодули* (*submodules*) для устранения проблем при обработке чрезвычайно больших файлов исходного кода с крупными модулями. Теперь к этому механизму добавилась возможность *импорта* (*import*), которая используется в интерфейсных блоках для импорта определений из внешнего модуля, что исключает необходимость определения второго модуля, включающего в себя только требуемые определения;
- ещё один шаг к уменьшению зависимости от оператора `GOTO`, – оператор `exit`, выполняющий переход к концу блока `if` или `select`;
- оператор `do concurrent` можно рассматривать как более гибкую альтернативу оператору/блоку `forall`. Он сообщает компилятору, что данный код может выполняться в параллельном режиме (в данном случае параллельное выполнение осуществляется в рамках одной и той же *копии*, а не между несколькими копиями, как при использовании комассивов);
- *внутренние процедуры* (*internal procedures*) теперь можно передавать как действительные аргументы, при этом внутренним процедурам доступны переменные в подпрограмме, где определены такие процедуры. Это упрощает решение некоторых проблем с интерфейсами (см. главу 5);
- в набор стандартных функций включены разнообразные функции Бесселя (Bessel functions) и функции запросов при работе с битами.

1.6. Что осталось неизменным

Все описанные выше новые функциональные возможности языка Fortran не повлияли на существующие программы, соответствующие старым стандартам. Исключение составили лишь те свойства, кото-